
y first  newsletter  for  2011  will,  hopefully,  make it 
out before the end of its allotted month, and it gives 
me my first opportunity to wish you all a happy and 

prosperous New Year!  I have another newsletter issue for you 
filled with SAS news, hints and tips that  I  hope you will find 
interesting.  In this issue of VIEWS News I would like to welcome 
our newest  VIEWS Consultant, Alexandra Riley, who, with our 
regular contributor LeRoy Bessler, expands on the multiple trend 
charts from the last issue.  There are also articles from our regular 
contributors Murphy Choy, Peter Crawford and Sanjay Matange, 
with  articles  on  calculating  value  on  risk  with  SAS/IML, 
determining  open  ODS destinations,  and  drawing Forest  plots 
with  PROC  SGPLOT.   There  is  another  regular  article  from 
Amadeus Software, David Shannon is giving you the chance to 
try  out  the  latest  Amadeus  Friday  Quiz.   Finally,  my regular 
Formats, Options, and Functions section has some new features in 
SAS 9.2.

M

I'm also pleased to inform you that I will be co-chairing the 
Reporting and Information Visualization section at SAS Global 
Forum in April, and LeRoy, Murphy, Sanjay, David and myself 
will  be  presenting  in  that  section.   Peter  will  be  chairing  the 
Coders' Corner section there too.  We will all be very pleased to 
talk to you there about VIEWS and/or SAS software.

If  you  would  like  to  contribute  an  article,  to  re-visit  and 
improve an existing article, or just discuss the possibility of doing 
so,  please  feel  free  to  send  an  email  to  me  at 
newsletter@views-uk.org.   A  list  of  subject  suggestions  for 
your articles can be found on the VIEWS web site, and anyone is 
very  welcome  to  add  to  that  list  by  sending  emails  to 
newsletter@views-uk.org with your own questions.

Philip R Holland (Newsletter Editor)

Membership of VIEWS is free and you can register for  
notification of all future VIEWS events by emailing us at  

membership@views-uk.org, making certain you include  
your name and postal address.  If you would like to receive  

email notification whenever a new issue of the free quarterly  
VIEWS News is published on the web site, please remember  

to also include your email address, and please inform us  
when you change it.

Did You Know?
If you have a useful hint or tip, please send it to the Editor,  
and share it with the VIEWS membership.

Forest Plot using PROC SGPLOT
[SAS 9.2] “A  Forest  Plot  is  a  graphical  display  designed  to 
illustrate  the  relative  strength  of  treatment  effects  in  multiple 
quantitative  scientific  studies  addressing  the  same  question” 
(Wikipedia).  While there are many different types of Forest Plots 
in use, this common version of the Forest Plot can be done using 
PROC SGPLOT with a few lines of code.

PROC SGPLOT  is  optimal  for  the  creation  of  single-cell 
graphs.  However, this procedure has the following features that 
can be leveraged to create this multi-cell looking graph:

This procedure supports two sets of axes: X, Y, X2 and Y2. 
These  axes can be used to effectively split  each axis into two 
regions.  This axis-split technique is used in the graph to draw the 
odds ratio on the left side using the X-Y axis, and the statistics 
columns on the right using the X2-Y axis.  The common Y axis 
ensures the two parts remain aligned with each other.

Each axis is designed to accumulate the data from multiple 
plot  statements  assigned  to  that  axis.   We  have  used  that 
technique to put multiple scatter statements on the X2 axis, one 
for each of the statistics.

The  MARKERCHAR=column option  is supported by the 
SCATTER statement.   This option displays the character  data 
from the column at the X, Y location for each observation.  We 
have used  this feature  to  display the statistics  columns on the 
right.

Data for the Forest Plot:

DATA work.forest;
 INPUT Study $1-16 grp OddsRatio LowerCL
       UpperCL Weight;
 FORMAT weight PERCENT5.
        Q1 Q3 4.2
        study2 $CHAR8.;
 ObsId = _N_;
 IF grp = 1 THEN DO;
  weight = weight * 0.05;
  Q1 = OddsRatio - OddsRatio * weight;     
  Q3 = OddsRatio + OddsRatio * weight;
 END;
 ELSE DO;
  study2 = study;
  study = "";
 END;
 DATALINES;
Modano  (1967)    1  0.590 0.096 3.634  1
Borodan (1981)    1  0.464 0.201 1.074  3.5
Leighton (1972)   1  0.394 0.076 2.055  2
Novak   (1992)    1  0.490 0.088 2.737  2
Stawer  (1998)    1  1.250 0.479 3.261  3
Truark   (2002)   1  0.129 0.027 0.605  2.5
Fayney   (2005)   1  0.313 0.054 1.805  2
Modano  (1969)    1  0.429 0.070 2.620  2
Soloway (2000)    1  0.718 0.237 2.179  3
Adams   (1999)    1  0.143 0.082 0.250  4
Truark2  (2002)   1  0.129 0.027 0.605  2.5
Fayney2  (2005)   1  0.313 0.054 1.805  2
Modano2 (1969)    1  0.429 0.070 2.620  2
Soloway2(2000)    1  0.718 0.237 2.179  3
Adams2   (1999)   1  0.143 0.082 0.250  4
Overall           2  0.328 0.233 0.462  .
RUN;
/*Reverse the order to keep Overall at bottom*/
PROC SORT DATA = work.forest;
 BY DESCENDING obsid;
RUN;
/*Populate LCL2 and UCL2 only for group=1 for
  display in graph.
  Add columns for OR, LCL, UCL and WT for
  display of the stat columns*/
DATA work.forest2;
 SET work.forest;
 FORMAT oddsratio lowercl uppercl 5.3;
 DROP grp;
 IF grp = 1 THEN DO;
  lcl2 = lowercl; 
  ucl2 = uppercl;
 END;
 or = 'OR';
 lcl = 'LCL';
 ucl = 'UCL';
 wt = 'Weight';
RUN;
ODS HTML CLOSE;
/*Compute Y axis offset percent based on number
  of observations in data*/
DATA _NULL_;
 pct = 0.75 / nobs;
 CALL SYMPUTX("pct", pct);
 CALL SYMPUTX("pct2", 2 * pct);
 SET FOREST NOBS = nobs;
RUN;

Create the style template:

/*Create custom template for Unicode text in
  footnote*/
PROC TEMPLATE;
 DEFINE STYLE styles.foreststyle;
  PARENT = styles.analysis;
  STYLE GraphFonts FROM GraphFonts / 
      'GraphFootnoteFont' =
      ("<MTsans-serif-unicode>", 10PT, ITALIC); 
 END;
RUN;

Create graph using PROC SGPLOT:

ODS ESCAPECHAR = '^';
ODS LISTING SGE = OFF IMAGE_DPI = 100
    STYLE = styles.foreststyle;
ODS GRAPHICS / RESET WIDTH = 7IN HEIGHT = 4.5IN
    IMAGENAME = "ForestPlot_PhilHolland";
TITLE "Impact of Treatment on Mortality by
 Study";
TITLE2 H = 8PT 'Odds Ratio and 95% CL';
FOOTNOTE J = l "This graph is created using 
SAS^{UNICODE '00AE'X} 9.2 SGPLOT Procedure";
PROC SGPLOT DATA = work.forest2 NOAUTOLEGEND; 
 SCATTER Y = study2 X = oddsratio /
     MARKERATTRS = GRAPHDATA2   
         (SYMBOL = DIAMONDFILLED SIZE = 10);
 SCATTER Y = study X = oddsratio /
     XERRORUPPER = ucl2 XERRORLOWER = lcl2
     MARKERATTRS = GRAPHDATA1
         (SYMBOL = SQUAREFILLED);
 SCATTER Y = study X = or /
     MARKERCHAR = oddsratio X2AXIS;
 SCATTER Y = study X = lcl /
     MARKERCHAR = lowercl X2AXIS;
 SCATTER Y = study X = ucl /
     MARKERCHAR = uppercl X2AXIS;
 SCATTER Y = study X = wt /
     MARKERCHAR = weight X2AXIS;
 REFLINE 1 100 / AXIS = X;
 REFLINE 0.01 0.1 10 /  AXIS = X
     LINEATTRS = (PATTERN = SHORTDASH)
     TRANSPARENCY = 0.5;
 INSET '               Favors Treatment' /
     POSITION = BOTTOMLEFT;
 INSET 'Favors Placebo' / POSITION = BOTTOM;
 XAXIS TYPE = LOG OFFSETMIN = 0
     OFFSETMAX = 0.35 MIN = 0.01 MAX = 100
     MINOR DISPLAY = (NOLABEL) ;
 X2AXIS OFFSETMIN = 0.7 OFFSETMAX = 0.05
     DISPLAY = (NOTICKS NOLABEL);
 YAXIS DISPLAY = (NOTICKS NOLABEL)
     OFFSETMIN = &pct2. OFFSETMAX = &pct.;
RUN;

Sanjay Matange, SAS Institute
This article is an excerpt from a forthcoming presentation,
“Tips and Tricks for Clinical Graphs” by Sanjay Matange

at SAS Global Forum 2011 in Las Vegas, NV, USA
.

All the Ways to Display Multiple Trend Charts 
Three Selected Examples and Their Code

[SAS 9.2] The article “All the Ways to Display Multiple Trend 
Charts”  in  VIEWS News issue 52  introduced  seven  ways  to 
display  multiple  trends  using  a  variety  of  SAS  graphical 
procedures.  It  also provided comments and comparison, which 
you might wish to review.  This article focuses on three selected 
examples and provides the relevant code.   The examples were 
generated  with SAS 9.2  and  produce  PNG and  HTML output 
files.

All of these examples take advantage of the ability in HTML 
to provide pop-up text to reliably provide the detail values of the 
data points.   Therefore,  numerous tick marks,  to help estimate 
(imprecisely) the detail of the data points, would be superfluous. 
Nevertheless, we do demonstrate the ability to optionally provide 
more  tick  marks  on  the  “traditional”  PROC  GPLOT  Overlay 
Chart.

In the code for all three examples, DESCRIPTION=' ' is used 
in  order  to  prevent  the  display  of  pop-up  text  showing  the 
SAS/GRAPH procedure used.  Such a pop-up is very distracting 
when trying to get  the pop-up text for the data points instead, 
which already requires mouse dexterity.

Example 1. SGPANEL LOESS Chart (graph)

%LET PATH = C:\VIEWS53;
DATA work.DowByDayIn1990
  (KEEP = Year Month MonthAbbrev MonthName Day
        Dow date);
 FORMAT Dow 5.;
 SET sashelp.citiday (KEEP = date snydjcm
    WHERE = (snydjcm NE .));
 Year = YEAR(date);
 IF Year EQ 1990;
 Month = MONTH(Date);
 MonthAbbrev = PUT(Date, MONNAME3.);
 MonthName = PUT(Date, MONNAME9.);
 Day = DAY(Date);
 Dow = ROUND(snydjcm, 1);
RUN;

DATA work.DowByDayIn1990_withAltText;
 LENGTH AltTextVar $200;
 SET work.DowByDayIn1990;
 AltTextVar = 'alt="On '
    || TRIM(LEFT(PUT(date, WEEKDATX.)))
    || '0D'X || 'the Dow was '
    || TRIM(LEFT(PUT(Dow, 5.))) || '"';
 /* '0D'X forces a line break */
RUN;
PROC MEANS DATA = work.DowByDayIn1990 
    MIN MAX RANGE NOPRINT;
 VAR Dow;
 OUTPUT OUT = MinMax MIN = DowMin MAX = DowMax 
         RANGE = DowRange;
RUN;
DATA _NULL_;
 SET MinMax;
 CALL SYMPUT('YMin',
    TRIM(LEFT(PUT(DowMin, 5.))));
 CALL SYMPUT('YMax',
    TRIM(LEFT(PUT(DowMax, 5.))));
 CALL SYMPUT('YRange',
    TRIM(LEFT(PUT(DowRange, 5.))));
RUN;
DATA _NULL_;
 RETAIN ActualYMax &YMax. ActualYMin &YMin.; 
 SET MinMax;
 CALL SYMPUT('YIncrement', 10);
 TentativeRoundedYMax = ROUND(ActualYMax, 10);
 IF ActualYMax LE TentativeRoundedYMax
    THEN RoundedYMax = TentativeRoundedYMax;
 ELSE RoundedYMax = TentativeRoundedYMax + 10;
 CALL SYMPUT('RoundedYMax',
    TRIM(LEFT(PUT(RoundedYMax, 5.))));
 TentativeRoundedYMin = ROUND(ActualYMin, 10);
 IF ActualYMin GE TentativeRoundedYMin
    THEN RoundedYMin = TentativeRoundedYMin;
 ELSE RoundedYMin = TentativeRoundedYMin - 10;
 CALL SYMPUT('RoundedYMin',
    TRIM(LEFT(PUT(RoundedYMin, 5.))));
RUN;
PROC FORMAT LIBRARY = work;
 VALUE MonthNm
  1 = 'January'
  2 = 'February'
  3 = 'March'
  4 = 'April'
  5 = 'May'
  6 = 'June'
  7 = 'July'
  8 = 'August'
  9 = 'September'
  10 = 'October'
  11 = 'November'
  12 = 'December';
RUN;
QUIT;

Example 1: SGPANEL LOESS Chart (code)

GOPTIONS RESET = ALL;
ODS NORESULTS;

ODS LISTING CLOSE;
ODS GRAPHICS ON / RESET = ALL IMAGEMAP = ON 

        IMAGENAME = 'SGPANEL_LOESS'
                  TIPMAX = 2500;
ODS HTML PATH = "&Path." (URL = NONE)
    BODY = 'SGPANEL_LOESS.HTML'
    (TITLE = 'SGPANEL LOESS Chart for Dow by  
    Day for Each Month in 1990');
TITLE 'SGPANEL LOESS Chart: Dow by Day for   
    Each Month in 1990';
PROC SGPANEL DATA = work.DowByDayIn1990
 DESCRIPTION=' ';
 PANELBY Month / COLUMNS = 4 ROWS = 3;
 LOESS Y = Dow X = Day / NOLEGFIT
    MARKERATTRS = (COLOR = BLUE SIZE = 8 

  SYMBOL = CIRCLEFILLED)
    LINEATTRS = (COLOR = RED PATTERN = 34);
 ROWAXIS LABEL = ' ' REFTICKS
    VALUES = (&YMin. &YMax.);
 COLAXIS LABEL = ' ' GRID
    VALUES = (1 8 15 22 31);
 FORMAT Month MonthNm.;
 FORMAT Dow 5.;
 FORMAT Day 2.;
RUN;
QUIT;
ODS HTML CLOSE;
ODS LISTING;

Example 2. SGPLOT LOESS Chart (with anti-aliasing)

GOPTIONS RESET = ALL;
ODS NORESULTS;
ODS LISTING CLOSE;
ODS GRAPHICS ON / RESET = ALL IMAGEMAP = ON 
        IMAGENAME = 'Anti-Aliased_SGPLOT_LOESS'
        TIPMAX = 2500 ANTIALIASMAX = 2500;
ODS HTML PATH = "&Path." (URL = NONE)
    BODY = 'Anti-Aliased_SGPLOT_LOESS.html'
    (TITLE = 'Anti-Aliased SGPLOT LOESS Chart 
    for Dow by Day for Each Month in 1990');
TITLE 'Anti-Aliased SGPLOT LOESS Chart: Dow 
    by Day for Each Month in 1990';

PROC SGPLOT DATA = work.DowByDayIn1990
 DESCRIPTION = ' ';
 LOESS Y = Dow X = Day /
    GROUP = Month
    MARKERATTRS = (SIZE = 7
                   SYMBOL=CIRCLEFILLED)
    LINEATTRS = (THICKNESS = 3 PATTERN = 1);
 YAXIS LABEL = ' ' VALUES = (&YMin. &YMax.) 
    REFTICKS;
 XAXIS LABEL = ' ' GRID
    VALUES=( 1  2  3  4  5  6  7  8  9 10
            11 12 13 14 15 16 17 18 19 20
            21 22 23 24 25 26 27 28 29 30 
            31);  
 FORMAT Month MonthNm.;
 FORMAT Dow 5.;
 FORMAT Day 2.;
RUN;
QUIT;
ODS HTML CLOSE;
ODS LISTING;

Example 3. “Traditional” PROC GPLOT Overlay Chart

GOPTIONS RESET = ALL;
ODS NORESULTS;
ODS LISTING CLOSE;
ODS HTML PATH = "&Path." (URL = NONE)
    BODY = 'HTML_GPLOT.html'
    (TITLE = 'HTML GPLOT Chart for Dow by Day 
    for Each Month in 1990');
GOPTIONS HTEXT = 2.25 PCT XPIXELS = 983  
    YPIXELS = 544;
TITLE1 HEIGHT = 1 PCT ' ' JUSTIFY = CENTER 
    HEIGHT = 4 PCT 'HTML GPLOT Chart: Dow by 
    Day during Each Month in 1990';
PROC GPLOT   
    DATA = work.DowByDayIn1990_withAltText;
 SYMBOL1 INTERPOL = JOIN FONT = 'WEBDINGS' 
    VALUE = '6E'X HEIGHT = 0.75 R = 12;
 AXIS1 LABEL = NONE STYLE = 0 MAJOR = NONE 
    MINOR = NONE
    ORDER = &RoundedYMin. TO &RoundedYMax.
      BY &YIncrement.;
 AXIS2 LABEL = NONE STYLE = 0 MAJOR = NONE 
    MINOR = NONE ORDER = 1 TO 31 BY 1;
 LEGEND1 LABEL = NONE SHAPE = SYMBOL(3, 0.75);

 PLOT Dow * Day = Month / NAME = 'HTML_GPLOT' 
    DESCRIPTION = ' '
    VAXIS = AXIS1 HAXIS = AXIS2 NOFRAME   
    AUTOVREF AUTOHREF LEGEND = LEGEND1
    DESCRIPTION = ' ' HTML = AltTextVar;
 PLOT2 Dow * Day = Month /
        VAXIS = AXIS1 NOLEGEND;
 FORMAT Month MonthNm.;
RUN;
QUIT;
ODS HTML CLOSE;
ODS LISTING;

LeRoy Bessler and Alexandra Riley

Delta Gamma VAR using SAS/IML
Value at risk (VAR) is one of the important models that is widely 
used in portfolio risk and market risk management.  VAR has no 
standardized  approach  and  is  essentially  a  conceptual  model. 
Several  approaches  have  been  proposed  and  adopted  in 
modelling  VAR.   The  simplest  approach  is  the  delta-gamma 
approach. Here we will examine VAR modelling using SAS/IML.

VAR is one of the fundamental concepts in risk management. 
It  is  one  of  the  simplest  models  that  is  used  to  estimate  the 
amount  of  loss  of  a  portfolio  in  a  1  in  20  negative  market 
scenario.  Typically, the model is focused on the upper bound of 
the distribution of the portfolio in order to establish the maximum 
upper loss of a portfolio given an event of a given probability of 
occurrence.  As VAR is a conceptual model, there is no standard 
approach  originally.   Several  approaches  have  been  suggested 
and  the delta-gamma model  is  one  of  them.  It  is  one  of  the 
simplest and easiest approaches which does not require extreme 
computing power.

The  delta-gamma  approach  basically  assumes  linear 
relationships between the various components of the portfolio, as 
well as a multivariate normality assumption.  By estimating the 
variance and covariance matrix, we will now be able to calculate 
the overall variance of the portfolio through the use of standard 
matrix calculations.  The toughest portion of the process is the 
variance-covariance  matrix estimation, which can be  extremely 
time  consuming  for  large  datasets.   To  begin  estimating  the 
covariance matrix, one must first create the mean.  This is a very 
simple step.

First we create a N x 1 matrix J with values of 1 in it.  We 
then multiply the transposed matrix with the matrix containing the 
data and divide by the number of rows.

Mean = J*(J`*Data)/(No of rows of Data)
The resultant mean matrix table contains a row of the mean 

value which can then be used to calculate the variance-covariance 
matrix.  To achieve that, we first create the deviation table. 

Dev = Data – Mean
After  obtaining  the  deviation  table,  we  then  calculate  the 

variance-covariance  matrix  by  multiplying  it  by  itself  and 
dividing by the number of rows-1.

Variance = (Dev’*Dev)/(No of rows of Data - 1)
Once  the  matrix  has  been  created,  it  is  then  possible  to 

calculate  the  lower  loss  limit  through  simple  matrix 

multiplications.   In  order  to  do  that,  a  weight  matrix  must be 
declared before submitting the macro.

The macro incorporates all the matrix manipulation steps by 
using SAS/IML.   The macro produces  the variance-covariance 
matrix as well as the mean of the portfolio,  and the upper and 
lower limits of the portfolio value in a 1 in 20 scenario.  Below is 
the screen shot of the output from the macro. 

There are several outputs, of which the lower limit is the most 
important, as it indicates the lowest value of the portfolio in a 1 in 
20 scenario. 

The macro, being SAS/IML based will require the SAS/IML 
license.  At the same time, given the memory required for matrix 
manipulations,  this macro is not suitable for use on large SAS 
datasets.   Larger  data sets will also cause the macro to take a 
much longer time.

SAS/IML provides SAS programmers with a simple and easy 
way to  construct  the  VAR model  without  having  to  resort  to 
complex programming.  At the same time, users can easily apply 
the VAR model on data to manage their portfolio risk. 

Murphy Choy, University College Dublin

Ask The Consultant
This  part  of  VIEWS  News is  where  you  can  get  your  
technical questions answered.  Send your questions to the  
Editor.

Which ODS destinations are open? 
Q: Can you find out which ODS destinations are open?

A: [SAS 9.2] Sometimes in a data or  reporting macro, I  would 
like to  know which ODS destinations are  open,  but  it  seemed 
difficult to uncover.  If I wish to create a macro that uses ODS to 
write to just one destination I need to know what destinations to 
suppress.   There  is  a  source  of  this information but  it  is  in  a 
dictionary table not a macro variable.   Here is  a macro which 
returns a list of open ODS destinations in a string that  can be 
assigned to a macro variable or used in a data step assignment.

%MACRO ODSdests(dlm = #) / des =
 'returns delimited list of ODS destinations';
 %LOCAL dsid destination rc fist dest;
 %LET dsid = %SYSFUNC(OPEN(sashelp.vdest, is));
 /* No leading ampersand with %SYSCALL */ 
 %SYSCALL SET(dsid); 

 %LET rc = %SYSFUNC(FETCH(&dsid.));
 %DO %WHILE(NOT &rc.);
  %LET dest = &fist.&destination.;
  %DO; &dest. %END;
  %LET rc = %SYSFUNC(FETCH(&dsid.));
  %LET fist = &dlm.;
 %END;  
 %LET rc = %SYSFUNC(CLOSE(&dsid.));
%MEND ODSdests;

An example of calling the macro:

ODS HTML(ID = demo1) FILE = '!temp\demo1.html';
ODS PDF(ID =  demo2) FILE = '!temp\demo2.pdf';
%PUT now we have >%odsdests(dlm=%STR(,))<;
ODS _ALL_ CLOSE;
%PUT now we have >%odsdests(dlm=%STR( ))<;

The  source  of  the  info  is  dictionary.destinations,  accessed 
through the SASHELP view VDEST.  Note that this macro does 
not cross any statement boundaries, and can, therefore, be used 
anywhere in a SAS program. 

Peter  Crawford,  Crawford  Software  Consultancy 
Limited

Friday Quiz from Amadeus Software
The Friday Quiz originated from the Technical staff at Amadeus 
Software.  Most Fridays we would share a recent problem that 
was encountered  (and  solved)  to  compare  the  approaches  that 
were used.  The enthusiasm for thrashing around programming 
ideas was such that we were encouraged to share the Friday Quiz 
with the general public. 

Each  quarter  we  will  publish  a  new  challenge  for  SAS 
Programmers to solve.  Quizzes vary in style, but are designed 
not to require more than a few minutes thought.  We encourage 
readers to share their solutions (anonymous or credited) and we 
will publish a selection at the end of each quarter.

We  hope  our  quizzes  serve  as  a  pleasant  distraction;  but 
mostly we hope you learn new skills and techniques by sharing 
each other's code.

A View to a Kill
Imagine you work in a department that collects a large volume of 
data each month.

To analyse the data and feed valuable insight back to your 
business, you run an analysis procedure which requires that all 
the data from the current year are available in a single table.

There  are  no  issues  in  the  first  few  months  of  the  year, 
however  your  business  is  growing  and  ever  more  data  are 
accumulated throughout the year.  Recently you have found the 
cumulative data set is too big for the disk you have available.

A  colleague  mentioned  that  an  Amadeus  Consultant 
(naturally)  had  recommended  considering  something  called  a 
SAS View.

The  following  program  illustrates  how  you  have  been 
appending the data together (albeit  a very small sample of the 
problem!).

/* Build the problem */
%MACRO month;
 %DO m = 1 %TO 12;
  DATA mon&m.;
   DO customer = 1 TO 20;
    OUTPUT;
   END;
  RUN;
 %END;
%MEND month;
%month
/* This works, but is really slow to build and 
   takes up too much disk space */
DATA all_months;
 SET mon:;
RUN;

Modify  the  data  step  so  that  ALL_MONTHS  requires 
negligible disk space.  Given that both a View and Table of the 
same name cannot exist in the same library, extend the code so 
that it defends itself against this problem.

Finally,  what are the advantages and disadvantages of your 
solution?

If you wish us to consider your solution for publication, send 
it to fq@amadeus.co.uk by the 31 May 2011 and state if you are 
happy for Amadeus to publish your name along with the code. 
Solutions can be obtained after that date by request from the same 
email  address,  or  by  visiting  the  Friday  Quiz  web  page 
www.amadeus.co.uk/fq.

David Shannon, Amadeus Software

Formats, Options, and Functions
This  section  is  devoted  to  the  description  of  useful,  or  
unusual, SAS Formats, Options, and Functions.

[SAS 9.2] The  ENVLEN() function  returns  the  length  of  an 
environment  variable  supplied  in  quotes.   If  the  environment 
variable does not exist, SAS returns -1.

[SAS 9.2] The  LRECL= option can now be used to specify 
record lengths on the Windows platform from 1 to 1,073,741,823 
for the following statements: FILE, FILENAME, %INCLUDE 
and INFILE.

[SAS 9.2] In  addition  to  writing  dates  in  the  form 
“ddmmmyy” or “ddmmmyyyy”, the DATEw. format now writes 
dates in the form “dd-mmm-yyyy”.

PhUSE News
2011 PhUSE Conference
The 2011 PhUSE conference is scheduled for 9-12 October 2011 
in  Brighton,  UK.   The  city  is  recognized  as  a  leading  and 
dynamic  conference  city  both  nationally  and  internationally, 
combining  a  seaside  location  with  a  vibrant  city  atmosphere. 
Brighton is under an hour by rail from London and thirty minutes 
from London Gatwick Airport, and, in the city itself, everything 
is within walking distance.

The  fantastic  venue  is  without  doubt  a  major  draw,  but 
ultimately  it  is  the  quality  of  the  conference  itself  that  will 

determine its success.  The Call for Papers (C4P) Tool is now 
open and  we urge  you  to  submit  a  paper  abstract  as  early as 
possible.  To help you to decide which stream you would prefer 
to present in, all stream descriptions are available on the PhUSE 
website,  including two new streams that  have been introduced 
this year.  Presenters are encouraged to consider the conference 
theme of 'Open Standards and Global Collaboration' when putting 
forward their papers.

We are again able to offer pre-conference training on Sunday 
8 October 2011.  This year we are pleased to announce that there 
will  be  two 'class  room'  based  training sessions on  'Advanced 
Enterprise Guide' and a 'CDISC Workshop', and there will also be 
the opportunity to become SAS Certified.

Several 'Discussion Clubs' will be run at the conference which 
will provide the ideal occasion to share knowledge and views and 
to  discuss  and  debate  experiences  for  a  number  of  topics 
including, standardisation and validation.  There will also be an 
opportunity  to  contribute  to  a  'live'  interactive  Wiki  where 
specific  content  will  be  discussed  whilst  the  Wiki  is  being 
updated online in real-time.

For  further  information,  please  see  the  PhUSE  website, 
www.phuse.eu.

PhUSE Committee

In Brief
• Back issues of VIEWS News are available from the VIEWS 

web site, and also from the VIEWS News page at 
www.sasCommunity.org/wiki/VIEWS_News

• Andrew Ratcliffe's has a blog at notecolon.info.
• More SAS hints and tips can be found on the SAS FAQ 

pages www.hollandnumerics.com/sasfaq, on the Amadeus 
Software Tips pages at www.amadeus.co.uk/tips and in 
their free podcasts at:
www.amadeus.co.uk/sas-technical-services/podcasts.

• There is a social networking group for SAS users on Ning 
called SAS Professionals, originally intended for SAS users 
in the UK, at www.sasprofessionals.net, but currently 
totalling over 4,400 members worldwide.

• Phil Holland has started a group page on Squidoo for other 
pages related to SAS, and would like to make it a central 
resource for SAS hints, tips, links and books at 
www.squidoo.com/groups/SAS_users

• Phil Holland has also created the SAS Contractors group on 
LinkedIn specifically for independent SAS consultants 
working in the UK and Europe.  Requests to join the group 
should be made at:
www.linkedin.com/groups?gid=2481512
and the associated SAS Contractor Resources page, which is 
open to all, can be found at: 
www.squidoo.com/sas_contractors
He also manages the SAS Professional Forum group on 
LinkedIn, which currently has over 8,700 members. 
Everyone is welcome to join, but discussions are restricted to 
SAS-related topics:
www.linkedin.com/groups?gid=70702

• SAS users who use Twitter can now join a “twibe” at 
twibes.com/SAS, which filters your tweets down to those 
containing “SAS”, “EnterpriseGuide” or “PROC”.

Diary
Are you organising an event that would be of interest to the  
VIEWS readership?  Let us know as we are interested in all  
non-profit making events related to SAS.

March 2011
8 PhUSE Single Day Event, Copenhagen, Denmark

See the PhUSE web site www.phuse.eu for further details.
15 SAS Programming Tips and Tricks for Base SAS, 

webinar
See the Amadeus web site www.amadeus.co.uk/webinars 

for further details.
24 Descriptive Statistics with Base SAS, webinar

See the Amadeus web site www.amadeus.co.uk/webinars 
for further details.
29 PhUSE Single Day Event, Basel, Switzerland

See the PhUSE web site www.phuse.eu for further details.

April 2011
4-7 SAS Global Forum, Las Vegas, NV, USA

The SAS Global  Forum (SGF, formerly called SUGI)  is 
held each year  and focuses on the technical aspects of SAS 
software.  See www.sasglobalforum.org for more details.
5 PhUSE Single Day Event, Paris, France

See the PhUSE web site www.phuse.eu for further details.
12 Using the SAS 9.2 Statistical Graphics Procedures, 

webinar
See the Amadeus web site www.amadeus.co.uk/webinars 

for further details.
28 PhUSE Single Day Event, Brussels, Belgium

See the PhUSE web site www.phuse.eu for further details.

May 2011
4 PhUSE Single Day Event, Frankfurt, Germany

See the PhUSE web site www.phuse.eu for further details.
11 PhUSE Single Day Event, London

See the PhUSE web site www.phuse.eu for further details.

June/July 2011
tba SAS Professionals Convention, Marlow

See the  www.sasprofessionals.net web site  for  details, 
when they become available.

October 2011
9-12 PhUSE Conference, Brighton

The  Pharmaceutical  Users  Software  Exchange  (PhUSE) 
conference  is  the  premier  programming  event  for  the 
pharmaceutical  industry in Europe.   See  www.phuse.eu for 
more details.

All VIEWS event information will be posted and/or emailed  
to registered members of VIEWS, and will include an event  
application  form.   Please  send  your  queries  about  any  
VIEWS events to  event@views-uk.org, and don’t forget to 
look at the web site for the latest news.

http://www.views-uk.org                                                              http://www.sascommunity.org/wiki/VIEWS_News

Rules can be viewed on our web site together with a list of  

